useful_identities.wxm

GNU General Public License 2019 Stephen Athel Abbott.

A short development document for Geometric Algebra with wxMaxima
taking its name from section 5 of the published paper...
A Survey of Geometric Algebra and Geometric Calculus (the Survey)
contains...
Initialization
Loading of functions (intrinsic and GA specific)
Pseudoscalar definition (specifies the space dimension) and
Calculation of the inverse pseudoscalar used to generate the dual of a multivector
Enumeration of the standard basis for the specified dimension

Check that the code is consistent with the identities at the end of 'the Survey'

Initialization

```
(%i59) reset()$
       kill(all)$
       startdisp:true$
       stringdisp:true$
       noundisp:true$
       simp:true$
       dotdistrib:true$
       derivabbrev:true$
       lispdisp:true$
```

load intrinsic (maxima or lisp) function files

```
(%i8)  load("basic")$
       load("facexp")$
       load("functs")$
       load("scifac")$
```

batchload GA specific (maxima) function files;

```
(%i12) ext:["wxm"]$
       file_type_maxima:append(ext,file_type_maxima)$

(%i14) batchload("gafns0")$
       batchload("gafns1")$
       batchload("gafns2")$
       batchload("gafns3")$
       batchload("gafns4")$
       batchload("gafns5")$
       batchload("gafns6")$
```

batchload GC specific (maxima) function files;

```
(%i21) batchload("gcfns1")$
       batchload("gcfns2")$
       batchload("gcfns3")$
```

the pseudoscalar and its inverse
the lowest useable dimension pseudoscalar should be {e1,e2} i.e. Plen = 2

```
(%i24) Pseudos:[e1,e2,e3]$
       Pvar:listofvars(Pseudos)$
       Plen:length(Pvar)$
       I:Pseudos$
       ni:(Plen-1)*Plen/2$
       Ii:(-1)^ni*I$
       kill(ni)$
       ldisplay(Pvar)$

 (%t31) Pvar=[e1,e2,e3]
```

initialize this list with the only list we have so far; Pvar

```
(%i32) lstbases:makelist(Pvar[n],n,1,Plen)$
```

the integer array, nbases (n:0,...3) is used for grader(M) in gafns4.wxm;
in order to use the same indices for the lists, (n:1,...3) define it using function array()

```
(%i33) array(nbases,Plen)$
       eset:setify(Pvar)$
       for ng:1 thru Plen do
       block(nbases[ng]:combination(Plen,ng),
       lstbases[ng]:full_listify(powerset(eset,ng)))$
       maxnbases:(combination(Plen,floor(Plen/2)))$
       nbases[0]:1$ /*an array index zero cannot be used to index any of the lists*/
       ldisplay(lstbases)$
       kill(eset,ng)$

 (%t38) lstbases=[[[e1],[e2],[e3]],[[e1,e2],[e1,e3],[e2,e3]],[[e1,e2,e3]]]
```

initialize this list again with Pvar

```
(%i40) lstblds:makelist(Pvar[n],n,1,Plen)$
```

the list named lstblds is used for grader(M) in gafns4.wxm;
lstblds is an list of lists of blades and allblds is a list of all blades

```
(%i41) for ng:1 thru Plen do
       block(lstb:lstbases[ng],
       lstblds[ng]:makelist(list2blade(lst),lst,lstb))$
       allblds:[]$
       for ng:1 thru Plen do
       block(allblds:append(allblds,lstblds[ng]))$
       ldisplay(lstblds)$
       ldisplay(allblds)$
       kill(lstb,ng)$

 (%t44) lstblds=[[[e1],[e2],[e3]],[[e1,e2],[e1,e3],[e2,e3]],[[e1,e2,e3]]]
 (%t45) allblds=[[e1],[e2],[e3],[e1,e2],[e1,e3],[e2,e3],[e1,e2,e3]]
```

end of Initialization

Some of the identities of Section 5 of 'the Survey' are grouped and separated from
the next identity by a blank line and they are un-numbered so we need to reference
the identity or group of identities with a number

firstly create three full grade multivectors

```
(%i47) nameA:"A"$
       makemultivec(nameA);
       nameB:"B"$
       makemultivec(nameB);
       nameC:"C"$
       makemultivec(nameC);

 (%o48) a_1,3*{e3}+a_2,3*{e2,e3}+a_1,2*{e2}+a_2,2*{e1,e3}+a_3,1*{e1,e2,e3}+a_2,1*{
e1,e2}+a_1,1*{e1}+a_0,1
 (%o50) b_1,3*{e3}+b_2,3*{e2,e3}+b_1,2*{e2}+b_2,2*{e1,e3}+b_3,1*{e1,e2,e3}+b_2,1*{
e1,e2}+b_1,1*{e1}+b_0,1
 (%o52) c_1,3*{e3}+c_2,3*{e2,e3}+c_1,2*{e2}+c_2,2*{e1,e3}+c_3,1*{e1,e2,e3}+c_2,1*{
e1,e2}+c_1,1*{e1}+c_0,1
```

the first identity is a strong test of the left inner product code;
it can take several seconds to run

```
(%i53) lhs:A&.(B&.C)$
       rhs:(A&^B)&.C$
       is(equal(lhs,rhs));

 (%o55) true
```

next using the full multivector, B;
for blades Ab(old) and Cb(old), when Ab is a subspace of Cb, Ab~^Cb=0;

```
(%i56) ldisplay(B)$
       Ab:a*{e1}$
       Cb:c*{e1,e2}$
       ldisplay(Ab,Cb)$
       subspacetest:Ab~^Cb$
       ldisplay(subspacetest)$
       lhs:(Ab&.B)&.Cb$
       rhs:Ab&^(B&.Cb)$
       ldisplay(lhs,rhs)$
       is(equal(lhs,rhs));

 (%t56) B=b_1,3*{e3}+b_2,3*{e2,e3}+b_1,2*{e2}+b_2,2*{e1,e3}+b_3,1*{e1,e2,e3}+b_2,1*
{e1,e2}+b_1,1*{e1}+b_0,1
 (%t59) Ab=a*{e1}
 (%t60) Cb=c*{e1,e2}
 (%t62) subspacetest=0
 (%t65)/R/ lhs=-c*b_2,1*a*{e1}+c*b_1,1*{e1,e2}*a
 (%t66)/R/ rhs=a*c*b_1,1*{e1,e2}-a*c*b_2,1*{e1}
 (%o67) true
```

miss out the third identity group

for the fourth identity, make four vectors (a, b, c, and d) to form a scalar identity

```
(%i68) lstga:[1]$
       namea:"a"$
       makelistgrademv(namea,lstga);
       lstgb:[1]$
       nameb:"b"$
       makelistgrademv(nameb,lstgb);
       lstgc:[1]$
       namec:"c"$
       makelistgrademv(namec,lstgc);
       lstgd:[1]$
       named:"d"$
       makelistgrademv(named,lstgd);
       lhs:(a&^b)&.(c&^d)$
       rhs:(a&.d)*(b&.c)-(a&.c)*(b&.d)$
       is(equal(lhs,rhs));

 (%o70) a_1,3*{e3}+a_1,2*{e2}+a_1,1*{e1}
 (%o73) b_1,3*{e3}+b_1,2*{e2}+b_1,1*{e1}
 (%o76) c_1,3*{e3}+c_1,2*{e2}+c_1,1*{e1}
 (%o79) d_1,3*{e3}+d_1,2*{e2}+d_1,1*{e1}
 (%o82) true
```

for the fifth identity group, just choose the second identity

```
(%i83) lhs:a&.(b&*c)$
       rhs:(a&.b)*c-(a&.c)*b$
       is(equal(lhs,rhs));

 (%o85) true
```

for the sixth identity group, just choose the second identity

```
(%i86) lhs:a&.(b&^c)$
       rhs:(a&.b)*c-(a&.c)*b$
       is(equal(lhs,rhs));

 (%o88) true
```